

Finish by Clipka

(February 2015)

Polished metals

Polished metals should only have a very low diffuse (if any) and minimum reflection should be 0.0. Use something like this:

```
texture {
  pigment {
    color rgbft <1.00, 0.62, 0.28, 0, 0>
  }
  finish {
    ...
    diffuse 0 // or some very low value
    metallic on
    ...
    reflection {
      0.722
      ...
      metallic on
    }
  }
}
```

Highlights

As far as specular highlights are concerned, a common misconception is that "specular 1.0" would correspond to a totally reflective surface - it does not, except for the hypothetical case of "roughness 0.0"; with the traditional "specular FLOAT" parametrisation, any increase in roughness also increases the total amount of reflected light. For a physically more straightforward parametrisation, use the new "specular albedo FLOAT" syntax, where "specular albedo 1.0" indeed corresponds to a totally reflective surface, regardless of the roughness. (Same goes for "phong FLOAT" vs. "phong albedo FLOAT" with respect to phong_size, and "diffuse FLOAT" vs. "diffuse albedo FLOAT" with respect to brilliance.)

Another thing that can really add to the credibility of a material is the simulation of fresnelian effects, not only for specular reflection (which gets brighter at shallow angles) but also for highlights (which should do the same) and diffuse reflection (which should get less bright at shallow angles). POV-Ray 3.7.0 only supports fresnelian effects for specular reflection (via the "fresnel on" statement in the reflection block), but not on highlights, and for diffuse reflections it only supports them when using subsurface scattering (in which case they actually can't be turned off). POV-Ray 3.7.1 adds full support for the latter two (by specifying "fresnel on" directly in the finish block as well; note that "fresnel on" in the reflection block is still required as well); UberPOV 1.37.1.X supports this new syntax as well.

Ideally, you would be using UberPOV 1.37.1.X with the following finish settings:

```
finish {
  diffuse albedo D brilliance B
  specular albedo S roughness R
  phong off
```

```
fresnel on
reflection { S roughness R fresnel on }
conserve_energy
subsurface { ... } // (optional)
}
```

where $D+S < 1$. (Don't forget to specify an interior block with `ior` to all your materials, even the opaque ones. If you don't know the refractive index of a material, 1.5 is a good bet.)

When using POV-Ray 3.7.1, for sufficiently shiny materials an acceptable approximation is to leave out the "roughness R" statement in the reflection block:

```
finish {
  diffuse albedo D brilliance B
  specular albedo S roughness R
  phong off
  fresnel on
  reflection { S fresnel on }
  subsurface { ... } // (optional)
}
```

For sufficiently dull materials, you can turn off reflections entirely:

```
finish {
  diffuse albedo D brilliance B
  specular albedo S roughness R
  phong off
  fresnel on
  reflection { 0.0 fresnel on }
  conserve_energy
  subsurface { ... } // (optional)
}
```

(Theoretically you would need to compensate by fudging D up a bit for the sake of radiosity but not for classic diffuse, but that's something POV-Ray doesn't support at present.)

For materials in between, there is no really good solution in official POV-Ray at present.

When using POV-Ray 3.7.0, you should fudge up the specular highlights and (unless you're using subsurface scattering) fudge down the diffuse reflection, to compensate for the fact that they don't honour fresnelian effects:

```
finish {
  diffuse albedo D*(1-F) brilliance B
  specular albedo S*F roughness R
  phong off
  reflection { S fresnel on }
  conserve_energy
}
```

or:

```
finish {  
  diffuse albedo D brilliance B  
  specular albedo S*F roughness R  
  phong off  
  reflection { S fresnel on }  
  conserve_energy  
  subsurface { ... } // (mandatory)  
}
```

where F is some fudge factor very roughly around 2.0.

High-quality diffuse inter-reflections are of high importance for credibility as well. If you're using POV-Ray, this means using high-quality radiosity settings; with UberPOV, I'd recommend using "no_cache" in the radiosity block instead.